# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: csslayer <wengxt AT gmail com>

pkgname=fcitx5-table-extra
pkgver=5.0.13
pkgrel=1
pkgdesc="Extra tables for Fcitx5"
arch=('any')
url="https://github.com/fcitx/fcitx5-table-extra"
license=('GPL')
depends=('fcitx5-chinese-addons')
makedepends=('extra-cmake-modules' 'boost')
source=("https://download.fcitx-im.org/fcitx5/$pkgname/$pkgname-$pkgver.tar.xz"{,.sig})
sha512sums=('d9e58777b85c1ab401d8d49ee7efe445964975ad352f262948af0ef38f2302e7a52c14480a3b9834e499a26e0f190818b1a9bc152865e5626fa8507deeb2133d'
            'SKIP')
validpgpkeys=('2CC8A0609AD2A479C65B6D5C8E8B898CBF2412F9') # Weng Xuetian <wengxt@gmail.com>

build(){
  cd $pkgname-$pkgver

  cmake -DCMAKE_INSTALL_PREFIX=/usr -DCMAKE_INSTALL_LIBDIR=/usr/lib .
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}
